
SMTP-Einstellungen

1. TCPIP V5.3

1.1 Konfiguration von SMTP 
1.2 SYS$STARTUP:SYLOGICALS.COM 
1.3 TCPIP$SMTP_COMMON 
1.4 Spaming-Filter 
1.5 User-Definitionen 
1.6 HTML-Mails 
1.7 Attachments 
1.8 IMAP 
1.9 SFF 

1.1 Konfiguration von SMTP

TCPIP> set config smtp /option=(eight,relay,top) 
(wichtig) 

file:///C|/Decus/tcpip.html (1 von 11) [08.04.2003 09:30:49]



SMTP-Einstellungen

TCPIP> set config smtp /subst=name="weichert.de" 
(überschreibt knotenname.domäne - nur für $MAIL) 
TCPIP> set config smtp /postmaster="Postmaster" 
(Probleme werden an Postmaster@weichert.de geschickt.) 

zum Aktivieren der Änderungen bei laufendem System: 
TCPIP> STOP MAIL 
TCPIP> DISABLE SERVICE SMTP 
TCPIP> ENABLE SERVICE SMTP 
TCPIP> START MAIL 

1.2 SYS$STARTUP:SYLOGICALS.COM 

$ DEFINE /SYSTEM/EXEC/NOLOG TCPIP$SMTP_COMMON user_disk:[tcpip$smtp] 
(Gemeinsames Verzeichnis für SMTP - wichtig für Cluster) 
$ DEFINE /SYSTEM/EXEC/NOLOG TCPIP$SMTP_JACKET_LOCAL 1 
(wichtig für POP3 - damit Mail-Kopf-Infos erhalten bleiben) 
$ DEFINE /SYSTEM/EXEC/NOLOG TCPIP$SMTP_8BITMIME_HACK 1 
(keine Umwandlung in 7-Bit für rausgehende Mails) 

$ DEFINE /SYSTEM/EXEC/NOLOG TCPIP$SMTP_POSTMASTER_ALIAS "Postmaster" 
(returned Mails: anstelle "TCPIP$SMTP@..." "Postmaster@...") 
$ DEFINE /SYSTEM/EXEC/NOLOG TCPIP$SMTP_PROHIBIT_USER_HEADERS 1 
(verbietet Usern, ihren Absender durch TCPIP$SMTP_FROM zu verändern) 
$ DEFINE /SYSTEM/EXEC/NOLOG MAIL$SYSTEM_FLAGS 19 
(16+2+1=19) 
1 - OpenVMS-Cluster - alle Knoten nehmen Mails entgegen 
2 - User erhält Notification auf allen Knoten 
4 - Zeit wird im Notification-String mit ausgegeben 
8 - DECnet Phase IV-Syntax wird für DECnet-Mails verwendet 
16 - DECnet-Plus-Syntax wird für DECnet-Mails verwendet 

1.3 TCPIP$SMTP_COMMON

Einfache Verteiler-Listen 

$ TYPE TCPIP$SMTP_COMMON:postmaster.dis 
weichert@weichert.de 
weichert@ruwel-pfullingen.de 
[EOF] 
Mails an Postmaster@weichert.de werden an alle Einträge dieser Liste geschickt. Das ist 
einfacher, als "Postmaster" unter TCPIP/SMTP zu definieren. 
Hier können auch andere Verteiler-Listen (*.DIS) stehen. 

Wichtig: In der UAF darf kein Account mit diesem Namen existieren. 

SMTP für mehrere Domainen konfigurieren: 

$ TYPE TCPIP$SMTP_COMMON:TCPIP$SMTP_LOCAL_ALIASES.TXT 
weichert.de 
ruwel-pfullingen.de 
[EOF] 

1.4 Spaming

TCPIP$SMTP_COMMON:SMTP.CONFIG (W:RE)

Relay-Zones: weichert.de, decus.de 
RBLs: 
Reject-Unbacktranslatable-IP: TRUE 
Security: Friendly 
Bad-Clients: *.*.*.* 
Good-Clients: 10.12.0.0/16 
Reject-Mail-From: reply@reply.com, chup65a@mailandnews.com 
Reject-Mail-From: newsmailbiz@yahoo.com 
Accept-Mail-From: MyFriend@Bekannt.com 

Relay-Zones: Nur für die eingetragenen Domainen arbeitet SMTP als Relay. 

Good-Clients: Alle "eigenen" Systeme. 

Standardmäßig gilt: 
Accept-Unqualified-Senders FALSE 
Accept-Unresolvable-Domains FALSE 

Eigene Texte können definiert werden, die beim rejecten der Mail verschicht werden. 
z.B. Reject-Mail-From-Text: Sie wurden als Spamer eingestuft, wir nehmen keine Mails 
mehr von Ihnen entgegen. 
SPAM-Relay-Text: Unser Mailsystem arbeitet unter OpenVMS. Spamer haben keine 
Chance. 

1.5 User-Einstellungen
LOGIN.COM 
DEFINE TCPIP$SMTP_FROM "Frank.Weichert@Ruwel-Pfullingen.de" 
Auf diese Weise kann der Absender der Mail verändert werden. 
z.B. automatisches Versenden von Mails aus BATCH-JOB's heraus 

Verhindern kann man das durch TCPIP$SMTP_PROHIBIT_USER_HEADERS 1 

Damit Mails an "Frank.Weichert" geschickt werden können: 
MAIL > SET FORWARD /USER="Frank.Weichert" WEICHERT 
MAIL > SHOW FORWARD /USER=* ! zum Ansehen 

$ MAIL text.txt /subj="Test-Mail" SMTP%"Postmaster@weichert.de" 

/noself - nicht an sich selbst, falls definiert 
/pers="Text" - wird an "From: " angehängt, aber nur, wenn TCPIP$SMTP_FROM nicht 
definiert ist. 

1.6 HTML-Mails

Text-Datei erstellen. Beispiel: 

Mime-version: 1.0
Content-Type: text/html;
charset="iso-8859-1"
Content-Transfer-Encoding: 8bit
         <------ leere Zeile, dient als Trennzeichen, WICHTIG
<HTML>
...
</HMTL>

 Versenden mit $ MAIL htmlfile.txt /subject ... (siehe oben)

1.7 Attachments

MIME-Attachments unter OpenVMS
$ HELP MIME
$ MIME :== $SYS$SYSTEM:MIME.EXE

 MAIL> EXTRACT/NOHEADER filename
 MIME> OPEN filename
 MIME> EXTRACT /ATTACH=2 DEMO.ZIP

 Mail mit Attachment erstellen
 MIME> NEW mail.txt
 MIME> ADD/CONTENT-TYPE="application/pdf" doku.pdf
 MIME> ADD/Content-Type="application/vms-backup" /Encoding="Base64" 
save.bck
 MIME> EXIT
 $ MAIL mail.txt /subj=... (siehe oben)

1.8 IMAP

Seit TCPIP V5.3 erstmalig dabei.
Vorteil gegenüber POP3: Mails bleiben unter VMS
Nicht alle Client-Funktionen können im Moment genutzt werden.
So funktioniert es gut:
 auf dem Client:
 - "Expunge" - Posteingang beim Beenden leeren
 - Ordner nur unter "Mail"
 - Kopien immer an Adresse verschicken, nicht in Ordner verschieben

 SYS$SYSDEVICE:[TCPIP$IMAP]TCPIP$IMAP.CONF
 Max-Connections:25
 Server-Port:143
 Upload-Supported: TRUE
 Server-Trace: TRUE

 TCPIP> SET SERVICE IMAP /LIMIT=1600
 (Siehe Release-Notes ECO 2)

Client-Ordner und MAIL-Folder
 "Posteingang" wird auf NEWMAIL in der MAIL.MAI gemappt
 "Mail" wird auf MAIL in der MAIL.MAI gemappt
 "Unterordner unter Mail" werden auf Folder in der MAIL.MAI gemappt

 Für alle anderen Ordner z.B. "Gesendete Nachrichten" legt IMAP 
zusätzliche *.MAI-Dateien an. Hier gibt es noch Probleme, deshalb nicht zu 

empfehlen.

 Wenn man vom Mail-Client aus Mails verschiebt oder löscht, dann werden 
diese Änderungen erst beim Beenden des Mail-Clients auf dem Server 
durchgeführt.
 IMAP startet bei Bedarf zusätzliche Prozesse. Geht der Bedarf zurück, 
dann "sterben" diese Prozesse. Clients verlieren dann die Verbindung. Erst 
nach Neustart des Clients ist die Verbindung zum IMAP-Server wieder da.
Problematisch ist das, wenn die gesendete Mail in diesem Zustand zusätzlich
in den SENT-Ordner auf dem IMAP-Server kopiert werden soll.

Obwohl noch nicht alles funktioniert, haben wir komplett auf IMAP 
umgestellt.
 Wichtig: TCPIP 5.3 ECO 2

1.9 SFF - Send from file

 UCX$SFF.EXE gibt es schon sehr lange im Paket MAJORDOMO
 Seit TCPIP V5.1 offiziell supported
 gestattet selbst "komponierte" Mails zu verschicken

 SFF :== $ SYS$SYSTEM:TCPIP$SMTP_SFF.EXE ! (SYLOGIN.COM)

 Beispiel:

MAIL FROM: <Frank.Weichert@Ruwel-Pfullingen.de>
RCPT TO: <Hugo.Mueller@Mueller.de>
DATA
Return-Path: Postmaster@Ruwel-Pfullingen.de
Message-ID: <OpenVMS-123456789@Ruwel-Pfullingen.de>
Date:
From: Frank Weichert <Frank.Weichert@Ruwel-Pfullingen.de>
MIME-Version: 1.0
To: Hugo Müller <Hugo.Mueller@Mueller.de>
Cc: Frank Weichert <Frank.Weichert@Ruwel-Pfullingen.de>
Subject: Test-Mail
Content-Type: multipart/mixed;
 boundary=""------------3F5D5BDC322221115DB6D12B"""
<CR>
--------------3F5D5BDC322221115DB6D12B
Content-Type: text/html; charset=iso-8859-1
Content-Transfer-Encoding: 8bit
<CR>
<HTML>
....

</HTML>
<CR>
--------------3F5D5BDC322221115DB6D12B
Content-Type: application/vnd.ms-excel;
 name="Test.xls"
 Content-Transfer-Encoding: 8bit
 Content-Disposition: filename="Test.xls"
<CR>
1<TAB>2<TAB>3<TAB>4<TAB>5<CR>
<CR>
--------------3F5D5BDC322221115DB6D12B
<CR>

 Mail versenden mit "$ SFF filename"

yahMAIL

2. yahMail

2.1 Allgemein 
2.2 Funktionen 
2.3 Nächste Features 

2.1 Allgemein
Freeware entwickelt von Mark Daniel (Australien) nur unter OpenVMS 
WEB-Interface zu OpenVMS $MAIL (verwendet MAIL$-Funktionen) 
kann NICHT gleichzeit mit POP3 verwendet werden 
kann gleichzeit mit IMAP verwendet werden 
Unterstützung unter APACHE (HP), OSU (David Jones), WASD (Mark Daniel) 
Aktuelle Version yahMAIL 1.71 
Quelle: www.tmk.com 
oder: ftp.process.com 

2.2 Funktionen

Public-Archive 
MIME-Unterstützung 
Deutsche Oberfläche und deutsche Anwenderbeschreibung 
Suchfunktionen 
Login-Funktionen des WEB-Servers wird verwendet (auch SSL) 
verwendet MAIL-Folder 
Verschieben, Kopieren, Löschen wird sofort und sehr schnell ausgeführt 
Ausdruck ohne Bilder 
Attachments werden nicht automatisch geöffnet 
Programm-Codes in der Mail werden nicht ausgeführt 
Refresh 

2.3 Nächste Features

Personal Spaming-Filter 
Zugriff auf "imap.MAI"-Dateien 
ab Version 2.x neues Layout 
- LDAP-Unterstützung 
- Weiterleiten mit Attachment 
- Antworten an "Alle" 

Majordomo

3. Ldap-Server

3.1 Allgemein 
3.2 Funktionen 
3.3 Install-Tips 

3.1 Allgemein

Zentrales Adress-Buch für eMail-Adressen 
und andere Informationen 

Freeware von HP 
befindet sich auf eBusiness-CD von OpenVMS 
X.500-OSI-Anwendung 
DECnet-Plus und OSAK müssen installiert sein 
DECnet-Plus - 'DECnet over IP' 

kompatibel mit Netscape und Outlook 

3.2 Funktionen

CHEMA-Definitionen für eMail-Adressbuch vorhanden 
$DXIM zum Verwalten unter DCL 
funktioniert im Cluster ohne Failover (Masterknoten) 
UTF-8-Zeichensatz (nur für 7Bit-Zeichen) 

3.3 Installations-Tips

$ set def dxd$directory 
$ mc ncl disable dsa 
$ mc ncl delete dsa 
$ dele dsa*.*.* 
$ dele dxd$dua*.*.* 
@sys$startup:dxd$dsa_configure 
$ mc ncl 
crea dsa 
crea dsa naming context "/C=De/O=Decus" 
ena dsa 
exit 
$ @sys$startup:dxd$dua_configure 
$ DXIM 
crea /C=DE/O=Decus attributes object=organization 
exit 

Eintrag mit DXIM anlegen 
$ DXIM 
crea /C=De/O=Decus/OU=Munich attr abject=organizationalUnit 
crea /C=De/O=Decus/OU=Munich/cn="Frank Weichert" attr - 
object=(organizationalPerson,person), sn="Weichert" 
set /C=De/O=Decus/OU=Munich/cn="Frank Weichert" attr l="Pfullingen" 
... Tel="+49(7121)700-340" 
... mail="Frank.Weichert@Ruwel-Pfullingen.de" 

Folgende X.500-Namen sind bereits definiert 
localityName 
postalCode 
initials 
physicalDeliveryOfficeName 
homePostalAddress 
postalAddress 
streetAddress 
otherfacsimileTelephoneNumber 
facsimileTelephoneNumber 
mobileTelephoneNumber 
homeTelephoneNumber 
otherTelephoneNumber 
department 
nickName=Nickname 
rfc822Mailbox 

otherMailbox 
URL 
givenName 
telephoneNumber 
st 
Title 
commonName 
surname 

Unter Netscape oder Outlook 
LDAP-Server - Stammverzeichnis: O=Decus, C=De 

Unter Netscape kann man zusätzlich die Darstellung und die Felder in der Datei prefs.js 
definieren. 
user_prefs("ldap_2.servers.Decus.attributes.custom1", "Url:URL"); 
user_prefs("ldap_2.servers.Decus.colums", "cn,mail,custom1"); 

4. Majordomo

4.1 Allgemein 
4.2 Funktionen 
4.3 Install-Tips 

4.1 Allgemein

Freeware entwickelt von Karol Zielonko (HP) 
befindet sich auf Freeware-CD von OpenVMS 
programmiert mit PERL (muß installiert sein) 
verwendet OpenVMS $MAIL 
und PIPE_MAILSHR.EXE und TCPIP$SMTP_SFF.EXE 

MAIL> SHOW FORWARD /USER=*
LISTE1         PIPE%"UCX$MAJORDOMO_HOME:RESEND.COM 
      UCX$MAJORDOM -l LISTE1 -r liste1@domain1.de liste1-OUTGOING"

 Beim Eintreffen einer Mail wird COM-Datei gestartet. 

4.2 Funktionen

die meisten gebräuchlichen Funktionen sind vorhanden 
- Liste erstellen 
- SUBSCRIBE/UNSUBSCRIBE 
- moderierte Listen 
- Public-Listen, Private Listen 
- HELP, WHO usw. 
- unterschiedliche Domainen 
- Limitierung Mailgröße 

4.3 Installations-Tips

UAF-Account 
UAF> copy tcpip$smtp ucx$majordom /uic=[3655,101] 
UAF> modif ucx$majordom /dir=[majordomo] 
UAF> modif ucx$majordom /defpriv=sysprv /priv=sysprv 
UAF> modif ucx$majordom /batch /nonetwork /pwdlife=none 
UAF> modif ucx$majordom /nopwdexp /flags=norestricted 

Mail-PIPE installieren - aus ...MAJORDOMO.PIPE] (CD) 

SYLOGIN.COM: 
sff :== $ SYS$SYSTEM:TCPIP$SMTP_SFF.EXE 

copy ucx$majordomo_startup.com SYS$STARTUP 
copy ucx$majordomo_systartup.template sys$startup:*.com 
Datei entsprechend anpassen: 'MAJDISKS' ersetzen 

@sys$startup:ucx$majordome_systartup 
@install 
@install_pipe_registry.com 

SYSTARTUP_VMS.COM: 
@SYS$STARTUP:MAILPIPE_STARTUP.COM 
@SYS$STARTUP:UCX$MAJORDOMO_STARTUP.COM 

Wichtig für Cluster: SYS$BATCH auf jedem Knoten 

Konfiguration im Majordomo-Login-Verzeichnis 
coppy majordomo.cf_template *.cf 
Editieren $whereami = "Weichert.De"; 
Crea/dir [.lists] /owner=ucx$majordom 
perl CREATE-MAJORDOMO-FWD (Eigentümer festlegen) 
perl CREATE-LIST (Liste erzeugen) 

Mail und More

5. Mail & More

5.1 Allgemein 
5.2 Out of Office 
5.3 "Postwurf" 
5.4 Beispiel 1 - Programm-Update 
5.5 Beispiel 2 - Mail & More 

5.1 Allgemein

Mail & More ist eine "Anleitung" 
- Software-Update auslösen bei Eintreffen einer Mail 
- SMS verschicken 
- Privater Spaming-Filter 
- VirenCheck 

verwendet PIPE_MAILSHARE.EXE 
- Mail kommt rein 
- wird weitergeleitet an 'PIPE' 

- Mail wird als Datei abgelegt 
- PIPE startet DCL-Prozedur mit P1='Dateiname der Mail' 

$ set sec/acl=(default,W:RE) sys$specific:[000000]tcpip$smtp.dir 
$ set file /prot=w:r TCPIP$SMTP_COMMON:TCPIP$SMTP_LOCAL_ALIASES.TXT 

5.2 Out of Office und Archivieren

Bei Abwesenheit leiten viele ihre Mails an eine andere Adresse weiter. 

MAIL> SET FORWARD "Fritz.Maier@Domain.de" 
Mail wird nur an Fritz Maier geschickt, selber bekommt man sie nicht. 

Wie kann ich die Mail weiterleiten und dennoch die Mail (wenn ich zurück bin) lesen? 
Wie kann ich Mails archivieren ? 

$TYPE TCPIP$SMTP_COMMON:WEICHERT_DIS.DIS 
_WEICHERT@Ruwel-Pfullingen.de 
MAIL1_BCK@Ruwel-Pfullingen.de 
Fritz.Maier@Ruwel-Pfullingen.de 

Wichtig: Unterstrich vor username, ansonsten looped die Mail 

MAIL> SET FORWARD /user=weichert weichert_dis@Ruwel-Pfullingen.de 

Backup Monats- oder Wochenwechsel 

$ DELETE USER_DISK:[MAIL2_BCK]*.*.*
$ DELETE USER_DISK:[000000]MAIL2_BCK.DIR.*
$ RENAME USER_DISK:[000000]MAIL1_BCK.DIR MAIL2_BCK.DIR
$ CREA/DIR USER_DISK:[MAIL1_BCK] /OWNER=MAIL1_BCK
$ OPEN /WRITE FILE NEU.TXT
$ WRITE FILE "Initialisiert"
$ CLOSE FILE
$ MAIL NEU.TXT /SUBJECT="Initialisiert" MAIL1_BCK

 Auto-Reply bei Out-Of-Office -> siehe Beispiel 2 "Mail & More"

"Postwurfsendung"

Ich möchte Mails regelmäßig an viele Adressen senden. 
Wie geht das? 

$TYPE TCPIP$SMTP_COMMON:DECUS_LUG.DIS 
Frank.Weichert@Ruwel-Pfullingen.de 
Fritz.Maier@Muster.de 
Holger.Fritsche@Demo-Center.de 

... 

Von jedem beliebigen Mail-Client irgendwo auf der Welt kann man nun eine Mail an 
DECUS_LUG@Ruwel-Pfullingen.de 
schicken und diese Mail wird an alle Adressen weitergeleitet. 

(Noch besser geht das mit MAJORDOMO.) 

5.4 Beispiel 1

Beispiel Software-Update Sophos Antivirus 

SYS$SYSTEM:PIPE_MAILSHR_REGISTRY.DAT 
*\*\MAIL_DISK:[SOPHOS]SOPHOS.COM 
MAIL> SET FORWARD PIPE%"MAIL_DISK:[SOPHOS]SOPHOS.COM user" 
'user' ist der VMS-Username, unter dem die COM-Datei gestartet werden soll. 

SOPHOS.COM 

$ set noon
$ alt=f$environment("DEFAULT")
$ set def nt_sweep
$ SPRICH="WRITE SYS$OUTPUT"
$ DELE ides.zip.*

$ DEFINE /USER SYS$OUTPUT IDES.ZIP
$ LYNX -dump http://www.sophos.com/downloads/ide/ides.zip
$ RENAME readides.txt *.old
$ ROLL:
$ OPEN FILE SAV.CFG /share
$ READ FILE SATZ
$ READ FILE SATZ
$ CLOSE FILE
$ ROLL = F$ELEMENT(1,"=",satz)
$ ROLLN = ROLL + 1
$ REQUEST "%%% SWEEP - AutoUpdate - SerialNumber=''ROLLN'"
$ WRITE:
$ OPEN /WRITE/ERROR=OPENERR FILE SAV.CFG
$ WRITE FILE "[Rollout]"
$ WRITE FILE "SerialNumber=''ROLLN'"
$ CLOSE FILE
$ REQUEST "%%% SWEEP - AutoUpdate - SerialNumber=''ROLLN' geschrieben"
$ DELE *.IDE.*
$ UNZIP "-o" IDES.ZIP
$ set def nt_sweep
$ set def [-.ide_update]
$ @start   ; Client-Update starten
$ EXIT
$ OPENERR:
$ WAIT 00:00:10
$ REQUEST "%%% SWEEP - AutoUpdate - SAV.CFG kann nicht geöffnet werden"

$ EXIT

5.5 Beispiel 2

Mail userdefiniert "behandeln" 

$ ! MAIL-AND-MORE.COM
$ ! -----------------
$ !
$ ! Autor: Frank Weichert (Frank.Weichert@Ruwel-Pfullingen.de)
$ ! 
$ ! Diese DCL-Prozedur bearbeitet Mails, bevor diese in den
$ ! NEWMAIL-Folder geschrieben werden.
$ !
$ ! Das kann verwendet werden für
$ ! - privater Spamfilter
$ ! - VirenCheck
$ ! - SMS/Pager-info
$ ! - Popup Windows-Info
$ ! - Autoreply "Out of Office"
$ ! - oder was Sie sonst noch mit Mails tun wollen
$ !

$ ! Wie funktioniert das ?
$ ! ----------------------
$ ! Der User muß in $MAIL auf den PIPE-Transport geforwardet werden.
$ ! Der Pipe-Transport submitted MAIL_and_MORE.COM mit P1 als Dateiname
$ ! unter dem VMS-User-Account in der SYS$BATCH.
$ ! Zuerst werden Viren- und Spamchecks durchgeführt.
$ ! Dann wird die Mail in den NEWMAIL-Folder geschrieben
$ ! unter Verwendung der "_username"-Syntax.
$ ! Anschließend können PopUP, Pager usw. ausgeführt werden.
$ !
$ ! Installation
$ ! --------------
$ ! 
$ ! 1. MAILPIPE muß installiert sein, damit PIPE-transport vorhanden ist.
$ !    MAILPIPE ist beim MAJORDOMO-Kit im Verzeichnis [.PIPE]. 
$ !
$ !    In SYS$SYSTEM:PIPE_MAILSHR_REGISTRY.DAT
$ !    folgender Eintrag hinzufügen (Namen entsprechend anpassen):
$ !    *\*\MAIL_DISK:[MAIL_AND_MORE]MAIL_AND_MORE.COM
$ !
$ !    MAILPIPE kann nur solche DCL-Prozeduren starten, die hier 
$ !    eingetragen sind.
$ !
$ ! 
$ ! 2. Um die Mail-Dateien PIPE*.tmp für alle User verfügbar zu machen:
$ !    $ set sec/acl=(default,W:RE) sys$specific:[000000]tcpip$smtp.dir

$ !    (In Cluster-Umgebungen auf jedem Knoten.)
$ !
$ ! 3. Falls TCPIP$SMTP_COMMON:TCPIP$SMTP_LOCAL_ALIASES.TXT existiert:
$ !    $ set file /prot=w:r 
TCPIP$SMTP_COMMON:TCPIP$SMTP_LOCAL_ALIASES.TXT
$ ! 
$ ! 4. Für die betreffenden User ('user' ersetzen):
$ !    MAIL> SET FORWARD 
PIPE%"MAIL_DISK:[MAIL_AND_MORE]MAIL_AND_MORE.COM user"
$ !    
$ !    Damit wird beim Eintreffen einer Mail
$ !    - die Mail in eine Datei  SYS$SPECIFIC:[TCPIP$SMTP]PIPE*.TMP geschrieben
$ !    - MAIL_AND_MORE.COM als User in der SYS$BATCH submittet mit
$ !      P1=Name der PIPE*.TMP-Datei (File, das die Mail enthält)
$ ! 
$ SET NOON
$ ! Mail-File ins User-Verzeichnis kopieren
$ copy 'p1 sys$login
$ !
$ !
$ ! ===== Behandlung vor dem Schreiben nach NEWMAIL =====
$ !
$ ! @mail_disk:[MAIL_AND_MORE]SPAM-CHECK P1
$ ! if $status .eqs. "error" then goto exit
$ ! @mail_disk:[MAIL_AND_MORE]VIRUS-CHECK P1

$ ! if $status .eqs. "error" then goto exit
$ ! ===== Ende =================================
$ !
$ ! ---- Schreiben der Mail nach NEWMAIL ----
$ !
$ infile=f$element(1,"]",p1)                    ! Copy der Original-Mail
$ outfile=f$element(0,".",infile) + ".out"      ! Neue Mail
$ !
$ ! Username mit "_" als erstes Zeichen
$ domain = f$trnlnm("TCPIP$BIND_DOMAIN")        ! IP-Domaine
$ uafuser = f$getjpi("","USERNAME")             ! UAF-Username
$ user = "_" + f$edit(uafuser,"COLLAPSE") + "@" + domain
$ !
$ header1 = ""
$ decnet  = ""
$ from = ""
$ cc = ""
$ subj = ""
$ !
$ open infile 'infile /read
$ open outfile 'outfile /write
$ !
$ read:
$ READ/END_OF_FILE=EOF infile satz
$ !
$ if header1 .eqs. "ok"     ! VMS-Mail-Header schon gelesen

$ then
$ WRITE outfile satz        ! Rest der Original-Mail satzweise anhängen
$ goto read
$ endif
$ !
$ ! Verarbeitung Mail-Header
$ !
$ if from .eqs. ""        ! From: einlesen
$  then
$  if f$locate("From:",satz) .ne. f$length(satz)
$   then
$   if f$locate("SMTP%",satz) .ne. f$length(satz)
$ !  ---- SMTP-Mail ----
$    then
$     from = f$element(1,"""",satz)
$    else
$ !  ---- Es ist eine DECnet-Mail -----
$     from = f$edit(satz,"COMPRESS")
$ !
$ !  generieren einer SMTP-like Adresse aus DECnet-From
$     from = f$element(1," ",from) + "@" + domain
$     decnet = "yes"
$   endif
$  endif
$ endif

$!
$! Betreff einlesen
$!
$ if subj .eqs. ""
$  then
$  if f$locate("Subj:",satz) .ne. f$length(satz)
$   then
$    subj = f$extract(5,f$length(satz),satz)
$    subj = f$edit(subj,"TRIM")
$  endif
$ endif
$ !
$! CC einlesen
$ if cc .eqs. ""
$  then
$  if f$locate("CC:",satz) .ne. f$length(satz)
$   then
$    cc = f$extract(3,f$length(satz),satz)
$    cc = f$edit(cc,"TRIM")
$  endif
$ endif
$ !
$ ! auf Ende des VMS-Mail-Header prüfen (leere Zeile)
$ !
$ if f$length(satz) .ne. 0 then goto read
$ !

$ header1 = "ok"     ! VMS-Mail-Header abgearbeitet
$ !
$ ! Die ersten Zeilen in der neuen Datei, erforderlich für SFF
$ !
$   write outfile "MAIL FROM:<" + from + ">"
$   write outfile "RCPT TO:<" + user + ">"
$   write outfile "DATA"
$ !
$ ! Wenn es eine DECnet-Mail war, dann schreiben wir jetzt den
$ ! RCF822-Header in die Datei
$ !
$   if decnet .eqs. "yes"
$    then
$     write outfile "Message-ID: <1234567890@" + domain
$     write outfile "Date: " + f$time()
$     write outfile "Cc: " + cc
$     write outfile "Subject: " + subj
$     write outfile "From: " + from
$     write outfile "Content-Type: text/plain; charset=iso-8859-1"
$     write outfile "Content-Transfer-Encoding: 8bit"
$     write outfile ""
$   endif
$ goto read
$ !
$ !

$ EOF:
$ close infile
$ close outfile
$ ! 
$ SFF 'outfile     ! schreibt die Mail nach NEMAIL
$ !
$ ! ====================== Nachbehandlung ========
$ !
$ GOSUB AUTOREPLY
$ GOSUB POPUP
$ GOSUB PAGER
$ ! ====================== Ende ===================
$! DELETE infile
$! DELETE outfile
$ EXIT
$ !
$ ! Nachbehandlungsroutinen
$ AUTOREPLY:
$ file=F$SEARCH("AUTOREPLY.TXT",1)
$ IF file .EQS."" THEN RETURN
$ MAIL 'file /SUBJ="Abwesenheits-Notiz" SMTP%"''from'" /NOSELF
$ RETURN
$ !
$ POPUP:
$! Pathworks muß installiert sein und PW-USER = UAF-USER
$ text = "eMail von: " + from + " Betreff: " + subj

$ ADMIN SEND 'uafuser "''text'"
$ RETURN
$!
$ PAGER:
$! Beispiel mit SMS-Sender der Firma ENZ-BERLIN (http://www.enz-berlin.de)
$! 
$ IF F$SEARCH("SMSNUMMER.TXT",1).EQS."" THEN RETURN
$ OPEN FILE SMSNUMMER.TXT
$ READ FILE SMS
$ CLOSE FILE
$ !
$ text = "eMail von: " + from + " Betreff: " + subj
$ OPEN FILE /WRITE/READ/SHARE=WRITE LTA5:
$ WRITE FILE "$''SMS',0,0,,,''text'"
$ CLOSE FILE
$ RETURN
$ !

